# $Id: PKGBUILD 82336 2013-01-15 11:35:08Z allan $
# Maintainer: Thorsten Töpper <atsutane-tu@freethoughts.de>
# Contributor: Alois Nespor <alois.nespor@gmail.com>
# Contributor: JerichoKru   <jerichokru(AT)htomail.com>

pkgname=pragha
pkgver=1.1.1
pkgrel=2
pkgdesc="A lightweight GTK+ music manager - fork of Consonance Music Manager."
arch=('i686' 'x86_64')
url="http://pragha.wikispaces.com/"
license=('GPL3')
depends=('libnotify' 'libcdio-paranoia' 'gstreamer0.10-base'
         'taglib' 'hicolor-icon-theme'  'gtk2'
         'dbus-glib' 'sqlite' 'desktop-file-utils'
         'libclastfm')
optdepends=('notification-daemon: OSD notification')
replaces=('dissonance')
install=pragha.install
source=("https://github.com/downloads/matiasdelellis/pragha/$pkgname-$pkgver.tar.bz2"
        cdio-includes.patch)
sha1sums=('a7b839df5799263f870ea65a7f272015e24637e9'
          '05a2e007800dc5395a0bb7f026ec7e388cd9488c')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  patch -Np1 -i "$srcdir/cdio-includes.patch"

  LIBS+="-ldbus-glib-1"  ./configure --prefix=/usr
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir/" install

  install -m 644 "data/$pkgname.desktop" "$pkgdir/usr/share/applications"
  install -d "$pkgdir/usr/share/pixmaps"
  install -m 644 "data/$pkgname.png" "$pkgdir/usr/share/pixmaps/"
  install -m 644 "data/$pkgname.1" "$pkgdir/usr/share/man/man1/"

}

# vim:ts=2:sw=2:expandtab

